

## 5. APPLICATIONS

```

read(8, ' (a)', end=400) name
f1 = test_(name)
write(6, ' (2(3x,a32)') name, dicname_
name=dicname_
f1 = test_(name)
write(6, ' (2(3x,a32)') name, tagname_

```

Fig. 5.4.7.1. Example application accessing aliased data names.

and DDL2 dictionaries, and, second, it links equivalent data names within the DDL2 dictionary. Aliasing also allows the use of synonyms appropriate to the application.

*CIFtbx* is capable of handling aliased data names transparently so that both the input CIF and the application software can use any of the equivalent aliased names. In addition, an output CIF may be written with the data names specified in the *CIFtbx* functions or with names that have been automatically converted to preferred dictionary names. If more than one dictionary is loaded, the first aliases have priority. We call the preferred dictionary name the 'root alias'.

The default behaviour of *CIFtbx* is to accept all combinations of aliases and to produce output CIFs with the exact names specified in the user calls. The interpretation of aliased data names is modified by setting the *logical variables* *alias\_* and *aliaso\_*. When *alias\_* is set to *.false.*, the automatic recognition and translation of aliases stops. When *aliaso\_* is set to *.true.*, the automatic conversion of user-supplied names to dictionary-preferred alias names in writing data to output CIFs is enabled. The preferred alias name is stored in the variable *dicname\_* following any invocation of a getting function, such as *numb\_* or *test\_*. If *alias\_* is set to *.false.*, *dicname\_* will correspond to the called name. The variable *tagname\_* is always set to the actual name used in an input CIF.

For example, the data name *\_atom\_site\_anisotrop.u[1][1]* in the DDL2 mmCIF dictionary is aliased to the data name *\_atom\_site\_aniso\_u\_11* in the DDL1 core CIF dictionary. In the example application of Fig. 5.4.7.1, showing the *CIFtbx* function *test\_run* with both the mmCIF and core dictionaries loaded, the specified data name *\_atom\_site\_aniso\_u\_11* is used to inquire as to the names used in an input CIF.

The execution of this code results in the following printout.

```

_atom_site_aniso_U_11
  _atom_site_anisotrop.u[1][1]
_atom_site_anisotrop.u[1][1]
  _atom_site_aniso_U_11

```

### 5.4.8. Implementation of the tools

Implementation of the *CIFtbx* tools is straightforward. The supplied source files in all versions are: *cifbx.f*, *cifbx.sys* (used in *cifbx.f*) and *cifbx.cmn* (used in local applications). More recent versions of *CIFtbx* (version 2.4 and later) require certain additional source files: *cifbx.cmf*, *cifbx.cmv*, *hash\_funcs.f* and *clearfp.f* (or *clearfp\_sun.f*).

The common file *cifbx.cmn* must be 'include'd into any local routines that use *CIFtbx* tools. The library in *cifbx.f* may be invoked by either (i) compiling and linking the resulting object file as an object library, or with explicit references in the application linking sequence (versions 2.4 and later require *hash\_funcs.o* as an additional object file); or (ii) including *cifbx.f* in the local application and compiling and linking it with the local program.

Approach (i) is more efficient, but for some applications approach (ii) may be simpler.

### 5.4.9. How to read CIF data

The *CIFtbx* approach to reading CIF data is illustrated using a simple example program *CIF\_IN* (Fig. 5.4.9.1), which reads the file *test.cif* (Fig. 5.4.9.2) and tests the input data items against the dictionary file *cif\_core.dic*. The resulting output is shown in Fig. 5.4.9.3.

The program *CIF\_IN* may be divided into the following steps, each tagged with the relevant reference letter in the comment records of the listing shown in Fig. 5.4.9.1.

*A:* Define the local variables. The *CIFtbx* variables are added with the line *include 'cifbx.cmn'*.

*B:* Assign device numbers to the files using the command *init\_*. The device number 1 refers to the input CIF, 3 to the scratch file and 6 (stdout) to the error-message files. The device number 2 refers to an output CIF, if we were to choose to write one.

*C:* Open a specific dictionary file named *cif\_core.dic* with the command *dict\_*. The code *valid* signals that the input data items are to be validated against the dictionary. In this application, *dict\_* is invoked in an *IF* statement that tests whether the command is successful.

*D:* Open the CIF *test.cif* with the command *ocif\_* and test that the file is opened.

*E:* Invoke the *data\_* command, containing a blank block code, to 'open' the next data block. The block name encountered is placed in the variable *block\_*, which in this application is printed.

*F:* Read the cell-length values and their standard uncertainties with the *numb\_* command, and print these out. Test whether all of the requested data items are found.

*G:* The *char\_* function is used to read a single *character string*.

*H:* The *name\_* function is used to get the data name of the next data item encountered.

*I:* This sequence illustrates how text lines are read. The *char\_* function is used to read each line and the *text\_* variable is tested to see whether another text line exists in this data item.

*J:* This sequence illustrates how a looped list of items is read. Individual items are read using *char\_* or *numb\_* functions and the existence of another packet of items is tested with the variable *loop\_*.

The resulting printout is shown in Fig. 5.4.9.3. In this figure, note the following:

(i) The first six lines of the printout are output by *CIFtbx* routines, not by the program *CIF\_IN*. They occur when the *data\_* command is executed and data items in the block *numbo\_jumbo* are read from the CIF and checked against the dictionary file. Note that this is when the *CIFtbx* routines store the pointers and attributes of all items in the data block. All subsequent commands use these pointers to access the data.

(ii) The '####' string in front of *\_cell\_length\_a* in the input CIF makes this line a comment and makes it inaccessible to *CIF\_IN*.

(iii) Data items may be read from a CIF in any order but looped items must normally be in the same list. If one needs to access looped items in different lists simultaneously, the *bkmark\_* command is used to preserve *CIFtbx* loop pointers.

### 5.4.10. How to write a CIF

Writing a CIF usually is simpler than reading an existing one. An example of a CIF-writing program is shown in Fig. 5.4.10.1. This example is intentionally trivial. The created CIF *test.new* is shown in Fig. 5.4.10.2. Note that command *dict\_* causes all output items to be checked against the dictionary *cif\_core.dic*. Unknown names are flagged in the output CIF with the comment

## 5.4. CIFTBX: FORTRAN TOOLS FOR MANIPULATING CIFS

```

                                PROGRAM CIF_IN
C
C....A.. Define the data variables
      include      'ciftbx.cmn'
      logical      f1,f2,f3
      character*32 name
      character*80 line
      real         cela,celb,celc,siga,sigb,sigc
      real         x,y,z,u,numb,sdev
      data         cela,celb,celc,siga,sigb,sigc /6*0.0/

C....B.. Assign the CIFtbx files
      f1 = init_( 1, 2, 3, 6 )

C....C.. Request dictionary validation check
      if(dict_('cif_core.dic','valid')) goto 100
      write(6,'(/a/)') ' Requested Core dictionary not present'

C....D.. Open the CIF to be accessed
100   name='test.cif'
      if(ocif_(name)) goto 120
      write(6,'(a///)') ' >>>>>>>> CIF cannot be opened'
      stop

C....E.. Assign the data block to be accessed
120   if(.not.data_(' ')) goto 200
      write(6,'(/a,a/)') ' Access items in data block ',bloc_

C....F.. Extract some cell dimensions; test all is OK
      f1 = numb_('_cell_length_a', cela, siga)
      f2 = numb_('_cell_length_b', celb, sigb)
      f3 = numb_('_cell_length_c', celc, sigc)
      if(.not.(f1.and.f2.and.f3)) write(6,'(a)') ' Cell lengths missing!'
      write(6,'(a,6f10.4)') ' Cell ',cela,celb,celc,siga,sigb,sigc

C....G.. Extract space group notation (expected char string)
      f1 = char_('_symmetry_cell_setting', name)
      write(6,'(a,a/)') ' Cell setting ',name(1:long_)

C....H.. Get the next name in the CIF and print it out
      f1 = name_(name)
      write(6,'(a,a/)') ' Next data name in CIF is ',name

C....I.. List the audit record (possible text line sequence)
      write(6,'(a)') ' Audit record'
140   f1 = char_('_audit_update_record', line)
      write(6,'(a)') line
      if(text_) goto 140

C....J.. Extract atom site data in a loop
      write(6,'(/a)') ' Atom sites'
160   f1 = char_('_atom_site_label', name)
      f2 = numb_('_atom_site_fract_x', x, sx)
      f2 = numb_('_atom_site_fract_y', y, sy)
      f2 = numb_('_atom_site_fract_z', z, sz)
      f3 = numb_('_atom_site_U_iso_or_equiv', u, su)
      write(6,'(1x,a4,4f8.4)') name,x,y,z,u
      if(loop_) goto 160

C
      goto 120
200   continue
      end

```

Fig. 5.4.9.1. Sample program *CIF\_IN*. See text for explanation.

'#< not in dictionary'. This applies to both looped and single data items.

A more complex example of writing a CIF is given in the program *cif2cif* available with the *CIFtbx* release. A similar program that reads a CIF and writes an XML file is *cif2xml*, also available with the *CIFtbx* release.

### 5.4.11. Error-message glossary

The *CIFtbx* routines will generate explicit error messages in the printout or in the created CIF if requested to do so (e.g. during dictionary checks). If data processing cannot continue (i.e. fatal

errors), an appropriate error message is placed in the printout and execution terminates. However, the default approach is to remain mute and for error detection to be monitored by the application program via the *CIFtbx* functions returning *.true.* or *.false.* values that tell the application program whether the command was performed correctly. This places the primary responsibility for error checking on the application software. The importance of this approach is that it enables the local application to respond to runtime problems in a controlled way and to take corrective action if it is possible. However, some types of processing errors, such as exceeding the dimensions of critical *CIFtbx* arrays, do require appropriate messages to be issued and for execution to cease.